// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/Presale.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "@openzeppelin/contracts/access/Ownable.sol";

interface IERC721 {
    function tokensOfOwner(
        address owner
    ) external view returns (uint256[] memory);
}

contract Presale is Ownable {
    IERC20 public BURNS;
    IERC20 public XBURNS;
    IERC721 public NFT;

    address payable public TREASURY;

    bool public contributeActive = false;
    bool public redeemActive = false;
    bool public refundActive = false;

    uint256 public minContribution;
    uint256 public rate;
    uint256 public precision;

    uint256 public totalContributions = 0;
    uint256 public totalBonus = 0;

    mapping(address => uint256) public contributions;
    mapping(address => uint256) public bonus;
    mapping(uint256 => bool) public redeemedNfts;
    mapping(address => bool) public redeemedTokens;
    mapping(address => bool) public refunded;

    constructor(
        address burns,
        address xburns,
        address nft,
        address payable treasury,
        uint256 minContribution_,
        uint256 rate_,
        uint256 precision_
    ) {
        BURNS = IERC20(burns);
        XBURNS = IERC20(xburns);
        NFT = IERC721(nft);

        TREASURY = treasury;

        minContribution = minContribution_;
        rate = rate_;
        precision = precision_;
    }

    function updateBurns(address value) public onlyOwner {
        BURNS = IERC20(value);
    }

    function updateXburns(address value) public onlyOwner {
        XBURNS = IERC20(value);
    }

    function updateNtf(address value) public onlyOwner {
        NFT = IERC721(value);
    }

    function updateTreasury(address payable value) public onlyOwner {
        TREASURY = value;
    }

    function updateContributeActive(bool value) public onlyOwner {
        contributeActive = value;
    }

    function updateRedeemActive(bool value) public onlyOwner {
        redeemActive = value;
    }

    function updateRefundActive(bool value) public onlyOwner {
        redeemActive = value;
    }

    function updateMinContribution(uint256 value) public onlyOwner {
        minContribution = value;
    }

    function updateRate(uint256 value) public onlyOwner {
        rate = value;
    }

    function updatePrecision(uint256 value) public onlyOwner {
        precision = value;
    }

    function withdrawErc20(address _token) external onlyOwner {
        if (IERC20(_token).balanceOf(address(this)) > 0) {
            IERC20(_token).transfer(
                TREASURY,
                IERC20(_token).balanceOf(address(this))
            );
        }
    }

    function withdrawEth() external onlyOwner {
        (bool sent, bytes memory data) = payable(TREASURY).call{
            value: address(this).balance
        }("");
        require(sent, "Failed to send Ether");
    }

    function userBoosterCount(address user) public view returns (uint256) {
        uint256[] memory nfts = NFT.tokensOfOwner(user);
        uint256 boosters = 0;
        for (uint256 i = 0; i < nfts.length; i++) {
            if (!redeemedNfts[nfts[i]]) {
                boosters++;
            }
        }
        return boosters;
    }

    function contribute() public payable {
        require(contributeActive, "CONT: Can't contribute yet.");
        uint256 amount = msg.value;
        require(amount >= minContribution, "CONT: Amount too low.");

        (bool sent, bytes memory data) = payable(TREASURY).call{value: amount}(
            ""
        );
        require(sent, "Failed to send Ether");

        contributions[_msgSender()] += amount;
        totalContributions += amount;

        uint256[] memory nfts = NFT.tokensOfOwner(_msgSender());
        uint256 boosters = 0;
        for (uint256 i = 0; i < nfts.length; i++) {
            if (!redeemedNfts[nfts[i]]) {
                redeemedNfts[nfts[i]] = true;
                boosters++;
            }
        }

        if (boosters > 0) {
            amount = (5 * boosters * amount) / 100;
            bonus[_msgSender()] += amount;
            totalBonus += amount;
        }
    }

    function redeem() public {
        require(redeemActive, "REDEEM: Can't redeem yet.");
        uint256 contribution = contributions[_msgSender()];
        require(contribution > 0, "REDEEM: No contribution.");
        require(!redeemedTokens[_msgSender()], "REDEEM: Already redeemed.");

        redeemedTokens[_msgSender()] = true;

        BURNS.transferFrom(
            TREASURY,
            _msgSender(),
            (contribution * rate) / precision
        );

        uint256 bonusContribution = bonus[_msgSender()];
        if (bonusContribution > 0)
            XBURNS.transferFrom(
                TREASURY,
                _msgSender(),
                (bonusContribution * rate) / precision
            );
    }
}